The Mathematica® Journal

Inversive Geometry

Part 2: Concyclic Points, Tangents, the Riemann Sphere, Rings of Four Circles and the Sierpinski Sieve

Jaime Rangel-Mondragón

This series of articles showcases a variety of applications around the theme of inversion, which constitutes a strategic way of manipulating configurations of circles and lines in the plane. This article includes the Riemann sphere, rings of four tangent circles, and inverting the Sierpinski sieve.

Introduction

The story of inversion, propelled by the study of conic sections and stereographic projection, is quite complex [1]. Although Vietá (1540–1603) already spoke of mutually inverted points, the proper development started with a plethora of notable geometers including L'Huilier (1750–1840), Dandelin (1794–1847), Quetelet (1796–1874), Steiner (1796–1863), Magnus (1790–1861), Plücker (1801–1868), Bellavitis (1803–1880), and Simpson (1849–1924). Steiner, in a manuscript published in 1913, is considered to have been the first to formulate inversion as a method to systematically simplify the study of com plex geometric figures where circles play a prominent role. This work culminated in 1855 with the studies of Möbius (1790–1868) (hence the choice of the letter *M* for the inversive circle in this article) [2]. Peaucellier (1832–1913) also applied inversion to his famous linkage [3], and Lord Kelvin (1824–1907) applied inversion to elasticity.

Let AB denote either the line or the segment from A to B, depending on the context. The length of the segment AB is denoted by $|\mathrm{AB}|$. The circle with center G and radius $\rho > 0$ is denoted by $\odot(\mathrm{G}, \rho)$. Denote by A' the inverse of A in the current circle of inversion $M = \odot(\gamma, \rho)$, where γ is either a symbolic point or often $(0, 0)$; in `Manipulate` results, *M* is often shown as a red dashed circle.

The following functions encapsulate the basic properties of inversive geometry developed in the first part of this series [4].

The function `squareDistance` computes the square of the Euclidean distance between two given points. (It is more convenient to use the following definition than the built-in Mathematica function `SquaredEuclideanDistance`.)

```
squareDistance[h_, k_] := Chop[(h - k).(h - k)]
```

The function `collinearQ` tests whether three given points are collinear. When exactly two of them are equal, it gives `True`, and when all three are equal, it gives `False`, because there is no unique line through them.

```
collinearQ[{a_, b_, c_}] :=
 (Chop[Det[Append[#, 1] & /@ {a, b, c}]] == 0)

collinearQ[{a_, a_, a_}] = False;
```

The function `circleABC[a, b, c]` computes the circle passing through the points *a*, *b*, and *c*. If the points are collinear, it gives the line through them; if all three points are the same, it returns an error message, as there is no meaningful definition of inversion in a circle of zero radius.

```
circleABC[{a_, a_, a_}] :=
 Return[
  "Three coincident points do not define a circle
    of positive radius."]

circleABC[{a_, b_, c_}] :=
 Line[Take[Union[{a, b, c}], 2]] /; collinearQ[{a, b, c}]

circleABC[{a_, b_, c_}] := Module[{center, x, y},
   center =
    First[
     {x, y} /.
      Quiet[If[And @@ NumericQ /@ Flatten[{a, b, c}],
          NSolve, Solve][squareDistance[{x, y}, a] ==
          squareDistance[{x, y}, b] ==
          squareDistance[{x, y}, c], {x, y}]]];
   Circle[center, Norm[center - a]]] /;
  Not@collinearQ[{a, b, c}]
```

The function `invert[M, p]` computes the inverse of p in the circle $M = \odot(\gamma, \rho)$. The object p can be a point (including the special point ∞ that inverts to the center γ of M), a circle, or a line (specified by two points).

```
invert[Circle[γ_, _], γ_] := ∞
```

```
invert[Circle[γ_, _], ∞] := γ
```

```
invert[Circle[γ_, ρ_], p : {_, _}] := γ + ρ^2 (p - γ) / squareDistance[p, γ]
```

```
invert[M : Circle[γ_, ρ_], Circle[cV_, rV_]] := Module[
  {n = Chop[squareDistance[cV, γ] - rV^2], h, k},
  If[n == 0,
   h = invert[M, cV + rV Normalize[cV - γ]];
   k = {{0, -1}, {1, 0}}.Normalize[cV - γ];
   InfiniteLine[{h + k, h - k}],
   Circle[γ + (cV - γ) ρ^2 / n, rV ρ^2 / Abs[n]]
  ]
 ]
```

```
invert[M : Circle[γ_, _], Line[{a_, b_}]] :=
 If[Chop[Det[{γ - a, γ - b}]] == 0,
  InfiniteLine[a, b],
  Chop[circleABC[{γ, invert[M, a], invert[M, b]}]]]
```

```
invert[M_, InfiniteLine[{a_, b_}]] := invert[M, Line[{a, b}]]
```

Here is an example.

```
invert[Circle[{2.5, 0}, 1.5], Line[{{5, 1}, {5, -1}}]]
```

```
Circle[{2.95, 0}, 0.45]
```

```
Module[{c, l},
 c = Circle[{2.5, 0}, 1.5];
 l = InfiniteLine[{{5, 1}, {5, -1}}];
 Graphics[{{Dashed, Red, c}, l, invert[c, l]}, Axes → True]
]
```



Since a circle can be inverted into a line, define a generalized circle to be either an ordinary circle or a line, as in the first article in this series. A consequence of theorem 2 in the next section is that the set of generalized circles is closed under inversion in a circle. The following `Manipulate` shows that as $r \to \infty$, the circle $M = \odot((-r, 0), r)$ tends to the line $x = 0$, and the inverse of the circle $\odot((-1, 0), 1/2)$ tends to its reflection in the line $x = 0$, the circle $\odot((1, 0), 1/2)$.

```
Manipulate[
 Module[
  {c, d},
  c = Circle[{-r, 0}, r];
  d = Circle[{-1, 0}, 1 / 2];
  Graphics[{
    {Dashed, Red, c},
    d, invert[c, d]
   }, PlotRange → 2, Axes → True]
 ],
 {{r, 20}, 1, 50, Appearance → "Labeled"},
 SaveDefinitions → True
]
```



Therefore, it makes sense to define inversion in a line to be reflection.

```
invert[Line[{w_, z_}], p : {_, _}] := Module[
  {nz = Normalize[z - w]},
  2 (w + nz.(p - w) nz) - p
 ]
```

```
invert[l : Line[_], Circle[γ_, ρ_]] := Circle[invert[l, γ], ρ]
```

```
invert[l : Line[_], Line[{p_, q_}]] :=
 Line[{invert[l, p], invert[l, q]}]
```

```
invert[l : InfiniteLine[_], x_] := invert[Line @@ l, x]
```

Here is an example.

```
Module[{l, c},
 l = InfiniteLine[{{0, 0}, {2, 1}}];
 c = Circle[{3, 1}, 1 / 3];
 Graphics[{{Red, Dashed, l}, c, invert[l, c]},
  Axes → True, ImageSize → 250]
]
```



The function `squareDistanceToLine` computes the square of the distance of a given point p to a given line.

```
squareDistanceToLine[p_, Line[{a_, b_}]] :=
 squareDistance[p - a,
  (p - a).(b - a) (b - a) / squareDistance[a, b]]
```

```
squareDistanceToLine[p_, InfiniteLine[{a_, b_}]] :=
 squareDistanceToLine[p, Line[{a, b}]]
```

The function `exCircles` computes the four circles tangent to the lines forming the sides of a triangle having given vertices.

```
exCircles[{a_, b_, c_}] := Module[{p, x, y, r2},
  p = {x, y};
  Circle[p, √r2] /.
   Quiet@NSolve[r2 == squareDistanceToLine[p, Line[{a, b}]] ==
      squareDistanceToLine[p, Line[{b, c}]] ==
      squareDistanceToLine[p, Line[{c, a}]], {x, y, r2}]]
```

The function `inCircle` computes the incircle of a triangle having given vertices as the excircle of smallest radius.

```
inCircle[{a_, b_, c_}] :=
 First[Sort[exCircles[{a, b, c}], Last[#1] < Last[#2] &]]
```

The function `redPoint` is used to mark a given point in red.

```
redPoint[p_, r_] := {EdgeForm[Thin], Red, Disk[p, r]}
```

The function `angle[a, b, c, r]` draws an arc of radius `r` and center `b` from the line `ab` to the line `bc`.

```
angle[a_, b_, c_, r_] := Module[{α, β},
  {α, β} = Sort[1. {ArcTan @@ (c - b), ArcTan @@ (a - b)}];
  If[β > α + π,
   {α, β} = {β, β + ArcCos[Normalize[a - b].Normalize[c - b]]}];
  {EdgeForm[Thin], Disk[b, r, {α, β}]}]
```

The function `intersections` computes the possible points of intersection of a given line and circle. (There may be zero, one, or two points.)

```
intersections[Line[{a_, b_}], Circle[q_, r_]] :=
 Module[{sol, x, y},
  sol = NSolve[{squareDistance[q, {x, y}] == r^2,
     Chop[Det[{{x, y} - a, b - a}]] == 0}, {x, y}, Reals];
  If[sol == {}, {}, {x, y} /. sol]]

intersections[InfiniteLine[{a_, b_}], Circle[q_, r_]] :=
 intersections[Line[{a, b}], Circle[q, r]]
```

The following `Manipulate` shows several of the functions introduced so far.

```
Manipulate[
 Module[{a, b, c},
  {a, b, c} = tri;
  Graphics[{ColorData[2, 6], EdgeForm[Thin],
    angle[b, a, c, 0.5], EdgeForm[Thick], ColorData[2, 5],
    Polygon[tri], Gray, InfiniteLine[{a, b}],
    InfiniteLine[{b, c}], InfiniteLine[{a, c}], Black,
    exCircles[tri], Blue, circleABC[tri], Red,
    redPoint[#, .08] & /@ tri},
   PlotRange → {{-5.2, 2.85}, {-5, 3.1}}, ImageSize → 400]
 ],
 "Drag the red points.",
 {{tri, 1. {{-1, -1}, {1, 0}, {0, 1}}}, Locator,
  Appearance → None}, SaveDefinitions → True
]
```



First Properties

The concept of orthogonality between circles plays an important role in inversive geometry. Two intersecting circles are orthogonal if they have perpendicular tangents at either point of intersection. It is rather amusing to consider orthogonality of circles without mentioning any notion of perpendicularity [3, 4], as is done in the following definition.

Definition

Two intersecting circles H and K are orthogonal if one of them (say H) belongs to a set of three mutually tangent circles that touch each other at three distinct points of K.

The following `Manipulate` shows the central circle is orthogonal to the black one by showing the other two circles (blue). Some values of the positions of the black circle or the radii of the blue circles make it impossible to close the three circles, so adjustments to close them are done on the fly.

```
Manipulate[
 Module[{rc, p, q, r, s, x, y, d, e, re},
  If[c.c < 1, c = 1.001 Normalize[c]];
  rc = Sqrt[c.c - 1];
  {p, q} =
   {x, y} /.
    Simplify[
     Quiet@
      NSolve[{x^2 + y^2 == 1, squareDistance[c, {x, y}] == rc^2},
       {x, y}]];
  d = p + rd Normalize[p - c];
  {r, s} =
   {x, y} /.
    Simplify[
     Quiet@
      NSolve[{x^2 + y^2 == 1, squareDistance[d, {x, y}] == rd^2},
       {x, y}]];
  {r} = Complement[{r, s}, {p, q},
    SameTest -> (Norm[#1 - #2] < 0.001 &)];
  re = Norm[q - r] /
     Norm[Normalize[r - d] - Normalize[q - c]];
  e = q + re Normalize[q - c];
  Graphics[{Circle[{0, 0}, 1], Circle[c, rc], Blue,
    Circle[d, rd], Circle[e, re], Red,
    Disk[#, 0.06] & /@ {p, q, r}},
   PlotRange -> {{-4, 4}, {-4, 4}}]],
```

```
{{c, {-2., 1}}, Locator},
{{rd, 1.5, "radius of blue circles"}, 0.01, 5,
 Appearance → "Labeled"},
Style["Drag the locator.", 10], SaveDefinitions → True
]
```



Theorems 1–11 review the basic properties of inversion and introduce some of its remarkable properties. Consider all inversions to be with respect to the circle $M = \odot(\gamma, \rho)$.

Theorem 1

Two pairs of inverse points are either collinear with γ or concyclic on a circle orthogonal to M.

The following `Manipulate` illustrates this property and shows that triangles QBA and QA'B' are similar. The gray circle is orthogonal to M.

```
Manipulate[
 Quiet@Module[{γ = {0, 0}, a, ad, b, bd},
   {a, b} = u;
   If[Chop[Norm[a]] < 0.001, a = 0.001 {1, 1}];
   {ad, bd} = invert[Circle[γ, r], #] & /@ {a, b};
   Graphics[{{Dashed, Red, Circle[γ, r]}, Black,
     Line[{γ, ad, bd, γ, a}], Line[{a, b}], Gray,
     circleABC[{a, b, bd}], ColorData[2, 5],
     {Opacity[.7], angle[γ, a, b, .2], angle[γ, bd, ad, .2],
      ColorData[2, 6], angle[γ, b, a, .15],
      angle[bd, ad, γ, .15]},
     Style[{Text["γ", γ, {0, 2.5}], Text["A", a, {3, -1}],
       Text["A'", ad, {-3, 0.7}], Text["B", b, {1, 2}],
       Text["B'", bd, {0, -3}]}, 15, Black],
     redPoint[#, 0.02] & /@ {γ, a, b, ad, bd}},
    PlotRange → {{-0.2, 1.65}, {-.7, .87}}, ImageSize → 400]
  ],
 {{r, 1, "radius of inversion"}, 0.01, 1.5,
  Appearance → "Labeled"},
 "Drag points A or B.",
 {{u, {{.6, .6}, {.55, -.2}}}, {-.02, -.7}, {1.65, .87},
  Locator, Appearance → None}, SaveDefinitions → True
]
```



The following interesting variation of theorem 1 was presented at the Swiss Mathematical Contest in 1999 [5, 6]:

> Two circles intersect in points D and E. An arbitrary point A of the first circle, which is not D or E, is joined to D and E, and the straight lines AD and AE intersect the second circle again in the points B and C. Prove that the tangent to the first circle at A is parallel to the straight line BC.

```
Manipulate[Module[{h, a, b, c, d, e, ok},
  h = (x^2 - r^2 + 1) / (2 x);
  ok = (Abs[h] < 1);
  If[ok, {d, e} = {{h, √(1 - h^2)}, {h, -√(1 - h^2)}}];
  a = {Cos[α], Sin[α]};
  ok = ok ⋀ (Chop[Norm[a - d]] > 0.001) ⋀
    (Chop[Norm[a - e]] > 0.001);
  If[ok,
   b = Select[intersections[Line[{a, d}], Circle[{x, 0}, r]],
     Norm[# - d] > 0.001 &];
   b = If[b == {}, d, First[b]];
   c = Select[intersections[Line[{a, e}], Circle[{x, 0}, r]],
     Norm[# - e] > 0.001 &];
   c = If[c == {}, e, First[c]]];
  Graphics[{Thick, Circle[], Circle[{x, 0}, r],
    If[ok, {Thin, Line[{a, d, b}], Line[{a, e, c}],
      Line[{d, e}], Line[{b, c}],
      redPoint[#, 0.05] & /@ {a, b, c, d, e},
      InfiniteLine[{a, a + b - c}],
      {ColorData[2, 6], Opacity[.7], angle[d, a, a + b - c, .4],
       angle[a, e, d, .4], angle[c, b, d, .4],
       ColorData[2, 5], angle[e, d, a, .4],
       angle[e, c, b, .4]},
      Style[{Text["A", a, {3, 0}], Text["B", b, {0, -2}],
        Text["C", c, {-4, 0}], Text["E", e, {0, 3}],
        Text["D", d, {0, -3}]}, 12, Black]}, {}]},
   PlotRange → {{-1.5, 4}, {-2.5, 2.5}}, ImageSize → 400]
 ],
 {{x, 2, "position of right circle"}, 0.01, 3,
  Appearance → "Labeled"},
 {{r, 1.5, "radius of right circle"}, 0, 2,
  Appearance → "Labeled"},
 {{α, 1. π, "position of point A"}, 0, 2. π,
  Appearance → "Labeled"}, SaveDefinitions → True
]
```



Theorem 2

The inverse of a circle not passing through γ is also a circle not passing through γ. (In this case, the images of concyclic points are concyclic.) The inverse of a circle passing through γ is a straight line not passing through γ. (In this case, the images of concyclic points are collinear.)

For instance, invert the circle $\odot((x, 0), r)$ in $M = \odot((0, 0), \rho)$.

```
Simplify[invert[Circle[{0, 0}, ρ], Circle[{x, 0}, r]],
 x > 0]
```

$$\begin{cases} \text{Circle}\left[\left\{\frac{x\,\rho^2}{-r^2+x^2},\ 0\right\},\ \frac{r\,\rho^2}{\text{Abs}\left[r^2-x^2\right]}\right] & r^2 \neq x^2 \\ \text{InfiniteLine}\left[\left\{\left\{\frac{\rho^2}{r+x},\ 1\right\},\ \left\{\frac{\rho^2}{r+x},\ -1\right\}\right\}\right] & \text{True} \end{cases}$$

The set of circles and lines, called *generalized circles*, is therefore closed under inversion.

Theorem 3

A straight line passing through γ is self-inverse. The inverse of a straight line not through γ is a circle through γ.

Theorem 4

Any circle through a pair of inverse points is orthogonal to the circle of inversion. Conversely, any circle cutting the circle of inversion orthogonally and passing through a point P *also passes through the inverse* P' [7].

Theorem 5

(The effect of inversion on angles.) The angle of intersection of two circles is the same as that of their inverse circles. This is also true for any intersecting curve; that is, inversion is conformal.

Theorem 6

(The effect of inversion on length.) If inversion sends point A *to* A' *and point* B *to* B'*, then*

$$\left| \mathrm{A'B'} \right| = \frac{|\mathrm{AB}|\, r^2}{|\gamma\mathrm{A}|\,|\gamma\mathrm{B}|},$$

$$\left| \mathrm{A'B'} \right| = \frac{|\mathrm{AB}|\, r^2}{|\gamma\mathrm{A}|\,|\gamma\mathrm{B}|}.$$

Proof

By the definition of inversion, $|\gamma\mathrm{A}||\gamma\mathrm{A'}| = |\gamma B||\gamma\mathrm{B'}| = r^2$. Then $|\mathrm{A'B'}|/|\mathrm{AB}| = |\gamma\mathrm{B'}| / |\gamma\mathrm{A}|$. Since $|\gamma\mathrm{B'}| = r^2/|\gamma\mathrm{B}|$, the result follows. □

Theorem 7

(The effect of inversion on the size of circles.) If the circle $\Gamma(\mathrm{G}, \alpha)$ *inverts into a circle of radius β, then*

$$\frac{\beta}{\alpha} = \frac{\rho^2}{|\alpha^2 - \Delta^2|},$$

where $\Delta = |\gamma\mathrm{G}|$*, the distance between the centers of M and* Γ.

To verify this result, notice that $\Delta^2 = (G-\gamma)\cdot(G-\gamma)$ in the following expression (the condition implies Γ does not pass through Q and thus Γ indeed inverts into a circle).

```
Last@FullSimplify[invert[Circle[γ, ρ], Circle[G, α]],
  α² ≠ (G - γ) . (G - γ)]
```

$$\frac{\alpha\,\rho^2}{\text{Abs}\left[\alpha^2 - (G-\gamma) . (G-\gamma)\right]}$$

Theorem 8

Two circles of radii α and β touching externally at the point γ invert into two parallel lines separated by the distance $\frac{\gamma^2}{2}\left(\frac{1}{\alpha}+\frac{1}{\beta}\right)$.

Subtracting the first points of the following lines obtained by inversion gives the result.

```
P1 = Simplify[invert[Circle[{x, 0}, ρ], Circle[{x - α, 0}, α]],
  α > 0]
```

$$\text{InfiniteLine}\left[\left\{\left\{x-\frac{\rho^2}{2\,\alpha},\,-1\right\},\,\left\{x-\frac{\rho^2}{2\,\alpha},\,1\right\}\right\}\right]$$

```
P2 = Simplify[invert[Circle[{x, 0}, ρ], Circle[{x + β, 0}, β]],
  β > 0]
```

$$\text{InfiniteLine}\left[\left\{\left\{x+\frac{\rho^2}{2\,\beta},\,1\right\},\,\left\{x+\frac{\rho^2}{2\,\beta},\,-1\right\}\right\}\right]$$

```
Simplify[P2[[1, 1, 1]] - P1[[1, 1, 1]]]
```

$$\frac{(\alpha+\beta)\,\rho^2}{2\,\alpha\,\beta}$$

Theorem 9

If the circle Δ of radius α inverts into a circle Δ' of radius α', then

$$\frac{\alpha'}{\alpha} = \frac{t^2}{\rho^2},$$

where t is the length of the tangent from γ to Δ' [8].

The following `Manipulate` shows the terms mentioned in theorem 9.

```
Manipulate[
 Module[{M = Circle[{0, 0}, 1], h, h1, a, b, ad, bd, β,
   x2, x3, y3, x4, y4},
  h = Circle[{pc, 0}, α];
  h1 = invert[M, h];
  β = h1[[-1]];
  x2 = h1[[1, 1]];
  {a, b, ad, bd} = {{pc - α, 0}, {pc + α, 0}, {x2 + β, 0},
    {x2 - β, 0}};
  x3 = (x2^2 - β^2) / x2;
  y3 = Sqrt[x2^2 - β^2 - x3^2];
  x4 = (pc^2 - α^2) / pc;
  y4 = Sqrt[pc^2 - α^2 - x4^2];
  Graphics[{{Thick, h, h1},
    redPoint[#, .025] & /@ {a, b, bd, ad}, Gray,
    Line[{{0, 0}, {3, 0}, {0, 0}, {x3, y3}, {x2, 0},
      {pc, 0}, {x4, y4}}],
    Style[{Text["γ", {0, -0.08}], Text["Δ", {pc, -α - 0.08}],
      Text["Δ'", {h1[[1, 1]], -β - 0.08}],
      Text[Style["t", Italic], ({x3 + x4, y3 + y4}) / 2,
       {0, -1}], Text["A", a, {1, 1.5}],
      Text["B", b, {-1.5, 1.5}], Text["B'", bd, {2, 1.5}],
      Text["A'", ad, {-2, 1.5}],
      Text["α", ({pc, 0} + {x4, y4}) / 2, {-2, 0}],
      Text["α'", ({x2, 0} + {x3, y3}) / 2, {-2, 0}]},
     12, Black], Red, Dashed, M},
   PlotRange → {{-.1, 3}, {-1, 1}}, ImageSize → 400]
 ],

 {{pc, 0.6, "position of circle Δ"}, 0.49, 1,
  Appearance → "Labeled"},
 {{α, 0.2, "radius of circle Δ"}, 0.1, 0.8,
  Appearance → "Labeled"}, SaveDefinitions → True
]
```



Proof

Let A and B be the ends of the diameter of Δ shown in the figure above. The point γ is outside the segment AB, as otherwise t would not be defined. Then

$$2\,\alpha = \left|\gamma B\right| - \left|\gamma A\right| = \frac{r^2}{|\gamma B'|} - \frac{r^2}{|\gamma A'|} = \frac{r^2(|\gamma A'| - |\gamma B'|)}{|\gamma A'|\,|\gamma B'|} = \frac{r^2\,2\,\alpha'}{|\gamma A'|\,|\gamma B'|},$$

and the result follows. □

For instance, consider inverting ⊙((1 / 2, 0), 1 / 3) in ⊙((0, 0), 2).

```
invert[Circle[{0, 0}, 2], Circle[{1 / 2, 0}, 1 / 3]]
```

$$\text{Circle}\left[\left\{\frac{72}{5}, 0\right\}, \frac{48}{5}\right]$$

Then $\beta = \frac{48}{5}$, $\alpha = \frac{1}{3}$, $t^2 = \left(\frac{72}{5}\right)^2 - \left(\frac{48}{5}\right)^2$, $r = 2$, which we verify as follows.

$$\frac{48/5}{1/3} == \frac{(72/5)^2 - (48/5)^2}{2^2}$$

```
True
```

In the case of theorem 7, this is the corresponding comparison.

```
48 / 5     2^2
------ == ----------------------------
1 / 3     Abs[(1 / 3)^2 - (1 / 2)^2]
```

```
True
```

In order to verify this result, assume that $\gamma = (0, 0)$ and $\Delta = \odot((x, 0), \alpha)$, with $x > 0$ and $x \neq \alpha$ (otherwise Δ' would not be a circle).

```
Module[{Δp, x, ρ, α, αp, t},
 Δp = Simplify[invert[Circle[{0, 0}, ρ], Circle[{x, 0}, α]],
   x > α > 0];
 αp = Last[Δp];
 t = Sqrt[Δp[[1,1]]^2 - αp^2];
 Simplify[αp / α == t^2 / ρ^2, x > α > 0]]
```

```
True
```

From this theorem, it follows that the product of the lengths of the tangents from γ to Δ and Δ' must be r^2.

Theorem 10

(Ptolemy) Let A, B, C, *and* D *be arbitrary points. Then* $|AC||BD| \leq |AB||CD| + |AD||BC|$, *with equality if and only if the points are on a generalized circle.*

Proof

Invert B, C, and D in the circle $\odot(A, \alpha)$. Then $|AB| = \frac{\alpha^2}{|AB'|}$ and $|CD| = \frac{\alpha^2}{|AC'||AD'|}|C'D'|$ (by theorem 6). Substituting this and similar results for AD, BC, AC, and BD in the inequality gives

$$\left(\frac{\alpha^2}{|AC'|}\right)\left(\frac{\alpha^2}{|AB'||AD'|}|B'D'|\right) \leq$$
$$\left(\frac{\alpha^2}{|AB'|}\right)\left(\frac{\alpha^2}{|AC'||AD'|}|C'D'|\right) + \left(\frac{\alpha^2}{|AD'|}\right)\left(\frac{\alpha^2}{|AB'||AC'|}|B'C'|\right),$$

which reduces to $|B'D'| \leq |C'D'| + |B'C'|$; this always applies in the triangle B'C'D'. Equality holds if and only if the points B', C', and D' are collinear, that is, if and only if points A, B, C, and D are concyclic or collinear. □

Ptolemy (circa 127 AD) compiled much of what today is the pseudoscience of astrology. His Earth-centered universe held sway for 1500 years. In the words of Carl Sagan, in the first episode of his TV series *Cosmos*, "… showing that intellectual brilliance is no guarantee against being dead wrong." You can find fascinating applications of Ptolemy's theorem in [9, 10].

Theorem 11

(A fixed-point theorem.) Suppose two nonintersecting circles have centers A *and* B. *Any circle that cuts those two circles orthogonally passes through the same two points on the line* AB [10].

This result is useful when inverting to concentric circles elsewhere. In order to verify it, assume without loss of generality that the two circles are $\odot((0, 0), \rho_1)$ and $\odot((u, 0), \rho_2)$, with $u > \rho_1 + \rho_2$, and that the orthogonal circle is $\odot((x, y), \rho_3)$. Then the following quantity is independent of y, and it is equal to the position of one of the two fixed points mentioned.

```
Module[{Q1, Q2, ρ3},
 With[{x = x, y = y, u = u, ρ1 = ρ1, ρ2 = ρ2},
  Q1 = {x, y};
  Q2 = {u, 0};
  First@FullSimplify[
    {x - Sqrt[ρ3^2 - y^2], 0} /.
     Solve[{Q1.Q1 == ρ1^2 + ρ3^2, (Q2 - Q1).(Q2 - Q1) == ρ2^2 + ρ3^2},
      {x, ρ3}], u > ρ1 + ρ2 > 0]
 ]
]
```

$$\left\{\frac{1}{2\,u}\left(u^2 + \rho 1^2 - \rho 2^2 - \sqrt{(u - \rho 1 - \rho 2)\,(u + \rho 1 - \rho 2)\,(u - \rho 1 + \rho 2)\,(u + \rho 1 + \rho 2)}\right),\ 0\right\}$$

■ The Riemann Sphere and Inversion

The following provides a framework with which to interpret inversion. Consider a sphere S^2 of unit radius centered at the origin. Draw a line l from the north pole N = (0, 0, 1) to a point P on the sphere. The point at which l intersects the x-y plane defines a one-to-one correspondence ξ between points on the sphere and points in the plane; ξ is called *stereographic projection*. The image of the south pole is the origin. The image of the north pole N is not defined, but ∞ is introduced as a new point to serve as the image of N; this makes the mapping ξ continuous and one-to-one. In the context of stereographic projection, the sphere is referred to as the *Riemann sphere* after the prominent mathematician Bernhard Riemann (1826–1866), who studied under Steiner and earned his PhD degree under Gauss [11, 12].

Similar triangles give

$$\xi(x, y, z) = \frac{(x, y)}{1 - z},$$
$$\xi^{-1}(x, y) = \frac{(2\,x, 2\,y, x^2 + y^2 - 1)}{1 + x^2 + y^2}.$$

By using these formulas, it is possible to prove that the stereographic projection of a circle on S^2 is a generalized circle in the plane. (Circles through N go to lines and other circles go to circles.) So inversion in the unit circle in the plane induces a map from circles to circles on S^2.

This defines stereographic projection.

```
ξ[{x_, y_}] := {2 x, 2 x, x^2 + y^2 - 1} / (1 + x^2 + y^2)
```

An inversive pair of points maps to points reflected in the x-y plane.

```
Module[{x, y, u, v},
 {u, v} = {ξ[{x, y}], ξ[invert[Circle[{0, 0}, 1], {x, y}]]};
Simplify[(Take[u, 2] == Take[v, 2]) ∧ (Last[u] == -Last[v])]]

True
```

Moreover, as the following `Manipulate` shows, an inversive pair of generalized circles maps to circles on S^2 that are reflections across the *x*-*y* plane. You can vary the radius and center of the control circle γ in the *x*-*y* plane. The circle γ inverts to δ in the unit circle, which is the equator of S^2. You can show the stereographic cone joining N to γ through its stereographic projection Γ and the stereographic cone joining N to δ through its stereographic projection Δ, which is the reflection of δ in the *x*-*y* plane [13].

```
Manipulate[
 Quiet@Module[
   {cx, cy, x, y, α, p1, p2, p3, u, r},
   If[Norm[cxy] > 1, cxy = Normalize[cxy]];
   If[Chop[Norm[cxy]] < 0.001, cxy = {0.01, 0.01}];
   {cx, cy} = cxy;
   Show[
    Graphics3D[{Opacity[op], Sphere[],
      Polygon[{{-5, -5, 0}, {5, -5, 0}, {5, 5, 0},
        {-5, 5, 0}}], Opacity[1], Point[{cx, cy, 0}],
      If[la, Style[{Text["γ", {cx, cy, 0}],
         Text["δ", invertTo3D[cxy]], Text["Γ", stereo[cxy]],
         Text["Δ", stereo[invert[cxy]]]}, 15], {}]},
     PlotRange → 2, Boxed → False, ImageSize → 400
    ],
    If[sc1,
     {p1, p2, p3} = {stereo[{cx + ra, cy}],
       stereo[{cx, cy + ra}], stereo[{cx - ra, cy}]};
     cone[Cross[p2 - p1, p3 - p1], .5 (p1 + p3), .5 Norm[p3 - p1],
      {0, 0, 1}],
     {}
    ],
    If[sc2,
     {p1, p2, p3} = {invert[{cx + ra, cy}],
       invert[{cx, cy + ra}], invert[{cx - ra, cy}]};
     {u, r} = List @@ circleABC[{p1, p2, p3}];
     cone[{0, 0, 1}, Join[u, {0}], r, {0, 0, 1}],
     {}
    ],
    ParametricPlot3D[x = cx + ra Cos[α];
     y = cy + ra Sin[α];
     {{x, y, 0}, {Cos[α], Sin[α], 0}, invertTo3D[{x, y}],
      stereo[{x, y}], stereo[invert[{x, y}]]}, {α, 0, 2 π}
    ],
    ViewAngle → 17 Degree
   ]
  ],
```

```
Row[{
   Spacer[60],
   Column[{
     Style["control circle γ   ", 9],
     Control@{{cxy, {0.5, 0}, "center"}, {-1, -1},
        {1, 1}},
     Control@{{ra, .3, "radius"}, 0, 1, ImageSize → Tiny}
    }, Alignment → Right],
   Spacer[30],
   Column[{
     Control@{{sc1, False, "γ-Γ cone"}, {True, False}},
     Control@{{sc2, False, "δ-Δ cone"}, {True, False}}
    }],
   Spacer[30],
   Column[
    {Control@{{op, .5, "opacity"}, 0, 1, ImageSize → Tiny},
     "",
     Control@{{la, False, "labels"}, {True, False}}
    }]
  }],
 Initialization → (
   invert[{0, 0}] = {0, 0};
   invert[p_] := p / (p.p);
   invertTo3D[p_] := Append[invert[p], 0];
   stereo[p : {x_, y_}] := {2 x, 2 y, -1 + p.p} / (1 + p.p);
   cone[v : {vx_, vy_, vz_}, c_, r_, f_] := Module[
     {a, b, λ, α},
     a = If[(vx == 0) ⋀ (vy == 0), {1, 0, 0},
       Normalize[{-vy, vx, 0}]];
     b = Normalize[Cross[a, v]];
     ParametricPlot3D[
      λ f + (1 - λ) c + r (1 - λ) (Cos[α] a + Sin[α] b),
      {α, 0, 2 π}, {λ, 0, 1}, Boxed → False, Axes → False,
      Mesh → False]
    ]), SaveDefinitions → True
]
```



Amusing applications related to stereographic projection are found in [14, 15], and interesting Demonstrations in [16–18].

A Ring of Four Tangent Circles

Consider a ring of four cyclically externally tangent circles as in the following `Manipulate`. The four points of tangency are concyclic (pink circle), even when some of the circles are internally tangent. You can drag the four points around this circle to alter the shape of the arrangement. To see many other similar patterns, see [19]. In contrast to the case of three tangent circles, the circle through the points of tangency is not necessarily orthogonal to the other four circles.

The function `perturb` slightly varies three points that are coincident or collinear.

```
perturb[{a_, b_, c_}] := If[
  Chop[Det[{b - a, c - a}]] != 0,
  {a, b, c},
  {a, b, c} + RandomReal[{-0.01, 0.01}, {3, 2}]
 ]

getCircle[V_, A_, B_] := Module[{v, a, b, center},
  {v, a, b} = perturb[{V, A, B}];
  center = LinearSolve[{v - a, v - b}, {a.(v - a), b.(v - b)}];
  Circle[center, Norm[center - a]]]

Manipulate[
 Module[{a, b, c, d},
  If[abcd ≠ Normalize /@ abcd, abcd = Normalize /@ abcd];
  {a, b, c, d} = abcd;
  If[a == b, b = a + .01];
  If[b == c, c = b + .01];
  If[c == d, d = c + .01];
  If[d == a, a = d - .01];
  Graphics[{EdgeForm[Thin], ColorData[2, 5],
    {Opacity[.5],
     Disk @@ getCircle[{0, 0}, a, b],
     Disk @@ getCircle[{0, 0}, b, c],
     Disk @@ getCircle[{0, 0}, c, d],
     Disk @@ getCircle[{0, 0}, a, d]
    },
    Pink, Circle[], Red, redPoint[#, 0.05] & /@ {a, b, c, d}},
   PlotRange → 3, ImageSize → 400]
 ],
 "Drag the red points.",
 {{abcd, {{0, -2}, {2, 0}, {0, 2}, {-2, 0}}}, Locator,
  Appearance → None}, SaveDefinitions → True
]
```



Let us apply inversion to deduce the concyclic property. Invert in a circle centered at one of the tangent points, for instance, the red dashed circle in the following `Manipulate`. That inversion transforms the circles centered at A and B into two parallel lines and two circles in between these lines. These four inversions are sequentially tangent in three points. The problem then reduces to show that the three points of tangency lie on the same line (shown in blue).

```
Manipulate[
 Module[{a, b, c, d, M, c1, c2, c3, c4, c11, c21, c31, c41},
  If[abcd ≠ Normalize /@ abcd, abcd = Normalize /@ abcd];
  {a, b, c, d} = abcd;
  M = Circle[b, r];
  {c1, c2, c3, c4} = {getCircle[{0, 0}, a, b],
    getCircle[{0, 0}, b, c], getCircle[{0, 0}, c, d],
    getCircle[{0, 0}, a, d]};
  {c11, c21, c31, c41} =
   Map[invert[M, #] &, {c1, c2, c3, c4}];
  Graphics[{
    {EdgeForm[Thin], ColorData[2, 5], Opacity[.5],
     Disk @@ c1,
     Disk @@ c2,
     Disk @@ c3,
     Disk @@ c4
    },
    Brown, Circle[],
    Blue, invert[M, Circle[{0, 0}, 1]],
    Black, {c11, c21, c31, c41},
    Red, Dashed, M,
    Black, Style[{
      Text["A", First[c1]],
      Text["B", First[c2]],
      Text["C", First[c3]],
      Text["D", First[c4]]
     }, 15],
    redPoint[#, 0.05] & /@ {a, b, c, d}
   }, PlotRange → 3, ImageSize → 400]
 ],
 {{abcd, {{0.5, -2}, {1.5, 0.5}, {0, 2}, {-2, 0}}},
  Locator, Appearance → None},
 {{r, 0.9, "radius of inversive circle"}, 0.01, 2,
  Appearance → "Labeled"},
 "",
 "Drag the red points.", SaveDefinitions → True
]
```



Let us take a closer look at the arrangement formed by the inversion of the chain of circles, starting from a different point of view. The following `Manipulate` shows an arrangement of two parallel lines, one slanted line, and two circles with the tangencies indicated. The two orange disks indicate adjustable tangency points A and B of the circles and the parallel lines. It is easy to show that the slanted line always passes through the tangency point of the circles regardless of the positions of A and B and the value of one of the radii, say, of the lower circle. In fact, the sum of the radii remains constant for fixed positions of A and B. For some values of the radii there exists a third circle also tangent to the parallel lines and the circles. Selecting the appropriate option in the `Manipulate`, the radii are adjusted according to the position of the slanted line for this third (blue) circle to exist. When the slanted line is vertical and the circles are congruent, three of those third circles exist.

```
Manipulate[
 Module[{h, r2, c},
  h = (1 + (up - lp)^2) / 2;
  If[r1 > h && Not@third, r1 = h];
  If[
   third,
   If[
    Abs[up - lp] < 0.005,
    If[
     Abs[r1 - 0.25] < 0.005,
     c = {up, up + 1 / √2 , up - 1 / √2 },
     c = {up}
    ],
    r1 = (2 h + Abs[up - lp] √(1 + 2 h) ) / 4;
    c = {(up + lp + Sign[up - lp] √(1 + 2 h) ) / 2}
   ]
  ];
  r2 = h - r1;
  Graphics[
   {{Thick, InfiniteLine[{{-1, 1}, {1, 1}}],
      InfiniteLine[{{-1, 0}, {1, 0}}]},
    InfiniteLine[{{lp, 0}, {up, 1}}], Circle[{lp, r1}, r1],
    Circle[{up, 1 - r2}, r2],
    If[third, {Blue, Circle[{#, 0.5}, 0.5] & /@ c}, {}],
    Disk[{up r1 + lp r2, r1} / h, 0.015], Green,
    Disk[{lp, 0}, 0.015], Disk[{up, 1}, 0.015]},
   PlotRange → {{-1, 1}, {-0.3, 1.3}}, ImageSize → 400]],
 "position of tangent point",
 {{up, 0.5, "upper"}, -1, 1, Appearance → "Labeled"},
 {{lp, 0, "lower"}, -1, 1, Appearance → "Labeled"},
 "",
 PaneSelector[
  {False → Control@{{r1, 0.2, "radius of lower circle"},
      0, 1, Enabled → Not@third, Appearance → "Labeled"},
   True → Spacer[20]}, third],
 {{third, False, "add third circle"}, {True, False}},
 SaveDefinitions → True
]
```



The next `Manipulate` inverts the previous arrangement. It shows that the tangent points of the set *S* of (blue) pairwise tangent circles are concyclic. It also shows that there exist up to three circles tangent to the circles of *S*, and sets of up to four circles orthogonal to the circles of *S*. You can vary the center of the inversive circle. Similar examples can be found at [20, 21].

```
Manipulate[
 Module[{h, r2, M, c, all, iall, all2, iall2, sl, alll, allc},
  If[Last[ic] == 1, ic = {First[ic], 1.001}];
  h = (1 + (up - lp)^2) / 2;
  If[third, If[Abs[up - lp] < 0.01,
    If[Abs[r1 - 0.25] < 0.01, c = {up, up + 1 / Sqrt[2], up - 1 / Sqrt[2]},
     c = {up}], r1 = (2 h + Abs[up - lp] Sqrt[1 + 2 h]) / 4;
    c = {(up + lp + Sign[up - lp] Sqrt[1 + 2 h]) / 2}]];
```

```
If[r1 > h, r1 = h];
 r2 = h - r1;
 M = Circle[ic, ric];
 sl = InfiniteLine[{{lp, 0}, {up, 1}}];
 alll = {InfiniteLine[{{-1, 1}, {1, 1}}],
    InfiniteLine[{{-1, 0}, {1, 0}}]};
 allc = {Circle[{lp, r1}, r1], Circle[{up, 1 - r2}, r2]};
 iall = Join[Map[invert[M, #] &, alll],
    Map[invert[M, #] &, allc]];
 If[third, all2 = Circle[{#, 0.5}, 0.5] & /@ c;
   iall2 = Map[invert[M, #] &, all2]];
 Graphics[{alll, allc, sl, invert[M, sl],
   {EdgeForm[Thin], Lighter[Blue, .7], Opacity[0.7],
    Disk @@@ iall}, If[third, {Blue, all2, Brown, iall2}, {}],
   iall, Disk[{up r1 + lp r2, r1} / h, 0.02], Green,
   Disk[{lp, 0}, 0.02], Disk[{up, 1}, 0.02], Red, Dashed, M},
  PlotRange → {{-2, 1}, {-1, 2}}, ImageSize → 400]
 ],
{{ic, {-1, 0.5}}, Locator},
Grid[{
  {"position of upper tangent point", Spacer[30], ""},
  {Control@{{up, 0.5, ""}, -2, 1, ImageSize → 150,
     Appearance → "Labeled"}, Spacer[30], ""},
  {"", "", ""},
  {"position of lower tangent point", Spacer[30],
   "radius of lower circle"},
  {Control@{{lp, 0, ""}, -2, 1, ImageSize → 150,
     Appearance → "Labeled"}, Spacer[30],
   Control@{{r1, 0.2, ""}, 0, 1, ImageSize → 150,
     Appearance → "Labeled"}},
  {"", "", ""},
  {"radius of inversion circle", Spacer[30],
   "add third circle(s)"},
  {Control@{{ric, 0.7, ""}, 0, 1, ImageSize → 150,
     Appearance → "Labeled"}, Spacer[30],
   Control@{{third, False, ""}, {True, False}}}
 }, Alignment → Left],
ControlPlacement → Top
]
```


position of upper tangent point
0.5
position of lower tangent point
0
radius of lower circle
0.5
radius of inversion circle
0.7
add third circle(s)

The quadrilateral joining the centers of the circles forming the ring has an inscribed circle. This is easily seen using Pitot's theorem (1695–1771): a convex quadrilateral with consecutive side lengths a, b, c, d is *tangential*, that is, it has an inscribed circle, if and only if $a + c = b + d$. Oddly enough, in the case of the quadrilateral, its incircle does not necessarily coincide with the circle passing through the four points of tangency, as the following result shows.

```
Module[{a, b, c, d, ab, bc, cd, ad, ra, r2, r3, r4, ci},
 {a, b, c, d} = {{-2.36, -2.33}, {1.11, -2.62},
   {1.14, 0.67}, {-2.36, 1.44}};
 {ab, bc, cd, ad} = Norm /@ {b - a, c - b, d - c, d - a};
 ra = 2.3;
 r2 = Abs[ab - ra];
 r3 = Abs[bc - r2];
 r4 = Abs[ad - ra];
 Chop[{Abs[r3 + r4 - cd], Abs[r3 - r4 - cd], Abs[r4 - r3 - cd]}];
 ci = circleABC[{(ra b + r2 a) / ab, (r2 c + r3 b) / bc,
    (r3 d + r4 c) / cd}];
 Graphics[{EdgeForm[Thin], ColorData[2, 5], Disk[a, ra],
   Disk[b, r2], Disk[c, r3], Disk[d, r4], Brown, ci,
   Blue, Line[{a, b, c, d, a}], Red,
   Circle[{-0.616, -0.726}, 1.74]}, ImageSize → 400]]
```



Inverting the Sierpinski Sieve

The *Sierpinski sieve* (or *Sierpinski gasket*, or *Sierpinski triangle*) [5, 6], named after the prolific Polish mathematician Waclaw Sierpinski (1882–1969), is a self-similar subdivision of a triangle [22, XXX]. The function `Sierpinski` constructs the n^{th} iteration corresponding to the recursive definition of this subdivision.

```
Sierpinski[n_] := Module[{s, a, b, c},
  Nest[
    (# /. s[{a_, b_, c_}] :> {s[{a, (a + b) / 2, (a + c) / 2}],
          s[{b, (b + c) / 2, (a + b) / 2}],
          s[{c, (a + c) / 2, (b + c) / 2}]}) &,
   s[Map[{Cos[#], Sin[#]} &, 2. π / 3 Range[3]]], n] /.
  s → Polygon]
```

```
Graphics[Sierpinski[5]]
```



Inverting the vertices of the triangles from the first four iterations in the unit circle gives rise to stunning patterns.

```
gr = GraphicsGrid[
  Partition[
   Table[
    Graphics[Sierpinski[n] /.
      Polygon[pts_] :>
       Polygon[invert[Circle[{0, 0}, 1], #] & /@ pts]],
     {n, 2, 5}], 2]]
```



```
gr /. Polygon → Line
```



The following `Manipulate` lets you vary the radius and center of the inverting circle, the number of iterations, and the type of transformation applied to the vertices forming the triangles forming the Sierpinski sieve.

```
Manipulate[
 Module[{Mm},
  Mm = Circle[γ, ρ];
  Graphics[{EdgeForm[Thin], {Red, Dashed, Mm},
    Sierpinski[n] /.
     Polygon[pts_] :→
      Switch[tr, 1, Polygon, 2, Line, 3, triad, 4, Mo, 5, Z][
       invert[Mm, #] & /@ pts]},
   PlotRange → {{-1.5, 1.5}, {-1, 1}},
   ImageSize → {450, 300}]],

 {{γ, {0.0, -0.01}}, Locator},
 {{ρ, 0.5, "radius of inversive circle"}, 0.2, 0.7,
  Appearance → "Labeled"},
 Row[{
   Control[{{n, 5, "iteration"}, Range[2, 6], Setter}],
   Spacer[200],
   Style["Drag the locator.", 10]
  }],
 {{tr, 2, "transformation"},
  {1 → "polygon", 2 → "line", 3 → "triad", 4 → "M",
   5 → "subtriangle"}, Setter},
 Initialization :→
  (triad[pts_] :=
    pts /. {a_, b_, c_} :→
      Line[{a, (a + b + c) / 3, b, (a + b + c) / 3, c}];
   Mo[pts_] :=
    pts /. {a_, b_, c_} :→
      Polygon[{a, (a + b) / 2, (a + c) / 2, (b + c) / 2, c}];
   Z[pts_] :=
    pts /. {a_, b_, c_} :→ {Line[{a, b, c, a}], Lighter@Blue,
       Polygon[{c + 2 a, b + 2 c, a + 2 b} / 3]}),
 SaveDefinitions → True
]
```



Acknowledgments

I would like to thank the anonymous referee whose thoughtful and detailed comments on this article greatly improved its presentation, and also Queretaro's Institute of Technology, which provided essential support for the completion of this second part.

References

[1] H. W. Eves, *A Survey of Geometry*, Boston: Allyn and Bacon, 1972.

[2] B. C. Patterson, "The Origins of the Geometric Principle of Inversion," *Isis*, **19**(1), 1933 pp. 154–180. www.jstor.org/stable/225190.

[3] J. Rangel-Mondragón. "Inversive Geometry II: The Peaucellier Inversor Mechanism" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/InversiveGeometryIIThePeaucellierInversorMechanism.

[4] J. Rangel-Mondragón, "Inversive Geometry: Part 1. Inverting Generalized Circles, Ellipses, Polygons, and Tilings," *The Mathematica Journal*, **15**, 2013. doi:10.3888/tmj.15-7.

[5] R. Todev, *Geometry Problems and Solutions from Mathematical Olympiads*, MathOlymps, 2010.

[6] D. Djukić, V. Janković, I. Matić, and N. Petrović, *The IMO Compendium*, 2nd ed., New York: Springer, 2011.

[7] D. E. Blair, *Inversion Theory and Conformal Mapping,* Providence, RI: The American Mathematical Society, 2000.

[8] H. Fukagawa and D. Pedoe, *Japanese Temple Geometry Problems: San Gaku*, Winnipeg, Canada: The Charles Babbage Research Centre, 1989.

[9] A. S. Posamentier, *Advanced Euclidean Geometry*, Emeryville, CA: Key College Publishing, 2002.

[10] E. W. Weisstein. "Ptolemy's Theorem" from Wolfram MathWorld—A Wolfram Web Resource. mathworld.wolfram.com/PtolemysTheorem.html.

[11] D. Pedoe, *Geometry, A Comprehensive Course*, New York: Dover Publications Inc., 1988.

[12] E. W. Weisstein. "Riemann Sphere" from Wolfram MathWorld—A Wolfram Web Resource. mathworld.wolfram.com/RiemannSphere.html.

[13] T. Needham, *Visual Complex Analysis*, Oxford: Claredon Press, 1997.

[14] D. Gehrig, "The Orloj," *The Mathematica Journal*, **7**(4), 2000. www.mathematica-journal.com/issue/v7i4/features/gehrig.

[15] P. W. Kuchel, "Spatial Inversion: Reflective Anamorphograms," *The Mathematical Journal*, **9**(2), 2004. www.mathematica-journal.com/issue/v9i2/SpatialInversion.html.

[16] J. Rangel-Mondragón. "A General Cone" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/AGeneralCone.

[17] P. W. Kuchel, "Anamorphoscopes: A Visual Aid for Circle Inversion," *Mathematical Gazette*, **63**(424), 1979 pp. 82–89.

[18] E. Mahieu. "Inverse Stereographic Projection of Simple Geometric Shapes" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/InverseStereographicProjectionOfSimpleGeometricShapes.

[19] A. Akopyan, *Geometry in Figures*, CreateSpace Independent Publishing Platform, 2011.

[20] J. Rangel-Mondragon. "Problems on Circles IV: Circles Tangent to Two Others with a Given Radius" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/ ProblemsOnCirclesIVCirclesTangentToTwoOthersWithAGivenRadius.

[21] J. Rangel-Mondragon. "Problems on Circles X: Tangent Circles Generate Ellipses" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/ProblemsOnCirclesXTangentCirclesGenerateEllipses.

[22] E. W. Weisstein. "Sierpiński Sieve" from MathWorld–A Wolfram Web Resource. mathworld.wolfram.com/SierpinskiSieve.html.

[23] A. Marquez-Raygoza. "Oftenpaper.net: The Sierpinski Triangle Page to End Most Sierpinski Triangle Pages." (Dec 5, 2016) www.oftenpaper.net/sierpinski.htm.

J. Rangel-Mondragón, "Inversive Geometry," *The Mathematica Journal*, 2016. dx.doi.org/doi:10.3888/tmj.18-5.

About the Author

Jaime Rangel-Mondragón received M.Sc. and Ph.D. degrees in Applied Mathematics and Computation from the School of Mathematics and Computer Science at the University College of North Wales in Bangor, UK. He was a visiting scholar at Wolfram Research, Inc. and held positions in the Faculty of Informatics at UCNW, the Center of Literary and Linguistic Studies at the College of Mexico, the Department of Electrical Engineering at the Center of Research and Advanced Studies, the Center of Computational Engineering (of which he was director) at the Monterrey Institute of Technology, the Department of Mechatronics at the Queretaro Institute of Technology, and the Autonomous University of Queretaro in Mexico, where he was a member of the Department of Informatics and in charge of the Academic Body of Algorithms, Computation, and Networks. His research included combinatorics, the theory of computing, computational geometry, and recreational mathematics. Jaime Rangel-Mondragón died in 2015.